Technical Report Series
Number 80-4

A COMPUTER PROGRAM FOR THE GRAPHIC DISPLAY AND QUANTITATIVE EVALUATION OF BEACH PROFILE DATA

C. F. Chamberlain
M. C. Larsen
and
G. F. Oertel

Georgia Marine Science Center
University System of Georgia
Skidaway Island, Georgia

TECHNICAL REPORT 80-4

A COMPUTER PROGRAM FOR THE GRAPHIC DISPLAY AND QUANTITATIVE EVALUATION OF BEACH PROFILE DATA

C.F. Chamberlain[1]

M.C. Larsen[2]

and

G.F. Oertel[3]

[1]Fugro, Inc.
3777 Long Beach Boulevard
Long Beach, CA 90807

[2]U.S. Geological Survey
Stop 99
Menlo Park, CA 94025

[3]Institute of Oceanography
Old Dominion University
Norfolk, VA 23508

The Technical Report Series of the Georgia Marine Science Center is issued by the Georgia Sea Grant Program and the Marine Extension Service of the University of Georgia on Skidaway Island (P.O. Box 13687, Savannah, GA 31406). It was established to provide dissemination of technical information and progress reports resulting from marine studies and investigations mainly by staff and faculty of the University System of Georgia. In addition, it is intended for the presentation of techniques and methods, reduced data and general information of interest to inudstry, local, regional and state governments and the public. Information contained in these reports is in the public domain. If this prepublication is cited, it should be cited as an unpublished manuscript.

ABSTRACT

A computer program for the plotting and evaluation of beach profile data provides a rapid solution to quantitative analytical problems. The program, written for a Hewlett-Packard desk top Calculator (Model 9825) - Plotter (Model 9862A), displays a graphic plot of beach face elevation data and quantifies the area under the beach profile. The data input required by the program includes: 1) an elevation at the profile origin, 2) the change in elevation measurements along the profile and 3) the distance between elevation measurements.

ACKNOWLEDGEMENTS

The development of this computer program was supported in part by a grant from the Georgia Office of Planning and Budget and the Georgia Sea Grant Program under NOAA,Office of Sea Grant #04-7-158-44126. This report is published as a part of the Georgia Marine Science Center's Technical Report Series. The authors appreciate the assistance of P.G. O'Malley and W.S. Chandler in developing the program. The authors thank Dr. V.J. Henry for reviewing the manuscript.

LIST OF FIGURES

INTRODUCTION

The collection of beach profile data has provided valuable information to coastal geologists and engineers. Measurements of changes in elevations across a beach face have been used for documenting changes in beach slope and shoreline positions, calculating sediment budgets and estimating littoral transport.

Qualitative and quantitative analysis of beach profile data has relied principally on graphic methods. The hand plotting of beach face elevations to obtain profiles and the use of a polar planimeter to evaluate the area under the profile is time consuming and invites operator error. A computer solution to this problem would seem obvious. Davis and Fox (1971) presented a computer program that maps erosion and deposition from analytical comparisons of topographic changes in a finite survey area. This paper presents a computer program that quantifies the area under the profile surface down to mean low water and eliminates operator error inherent in time consuming graphic evaluation methods. The program, designed for use with a Hewlett-Packard desk top calculator (Model 9825A) and plotter (Model 9862A), yields a graphic plot of the profiled surface, an areal calculation (in square meters) of the shore sediment prism, and a distance value (in meters) to mean low water from the profile origin.

The data input required by the program includes: an initial elevation (mean low water datum) at the profile origin, the sighting interval or spacing between elevation measurements, and the change in elevation measurements between profile sightings. All values are entered in centimeters.

PROGRAM DESCRIPTION

The program (Fig. 1) consists of graphic plot and calculation commands.

The graphic portion of the program displays the profile and plots the appropriate date, profile number, profile area, number of data points and distance from the initial elevation to mean low water (Fig. 2).

The arithmetic portion of the program determines the area under the profile to Mean Low Water by summing the areas under each sighting interval (S), where (E) is the change in elevation and (I) is the initial elevation of the next sighting (Fig. 3). The area under each sighting interval is evaluated by summing the areas of the rectangular and triangular segments.

The program is described by operations and statements. An operation is a procedure performed by the computer operator. A statement is a computer statement, calculation or plotter command.

PROGRAM USE AND OPERATION

Operation One

The data cartridge containing the program is inserted into the computer and rewound. The program is loaded into the core of the computer with the system command,"load," and the appropriate file number. Paper is placed on the plotter and the limits of the plotting area defined with the limit controls. The plotting limits determine the vertical exaggeration of the profile. A vertical exaggeration of 10 is established for a plotting area of 9.35 cm by 23.85 cm.

Operation Two

The operator keys "run" on the alphanumeric keyboard to begin the program.

```
0: scl -2000,
 18000,-100,550
1: axe 0,0,2000,
 100
2: csiz 3,1.5,.6
3: plt 6000,-50,
 1;lbl "20 METER
  INTERVALS"
4: csiz 1.3,.5,
 1,90
5: plt -1000,75,
 1;lbl "1 METER
 INTERVALS"
6: csiz 3,1.5,.6
7: plt 16500,20,
 1;lbl "MLW"
8: ent "Initial
 Elevation ?",I
9: 0→X;0→A;0→C
10: plt X,I
11: ent "Sightin
 g Interval in
 cm.?",S;S/2→B
12: ent "Change
 in elevation?",
 E
13: I+E→I;C+1→C
14: X+S→X
15: plt X,I
16: if I<0;sto
 "comp"
17: I*S→R;abs(E)
 *B→T
18: R+T→Q;Q/1000
 0→Q
19: A+Q→A
20: sto 12
21: "comp":I-
 E→I;X-S→X
22: I*S/abs(E)→Z
23: plt X+Z,0,1
24: Z/2*I→Q
25: Q/10000→Q
26: A+Q→A
27: (X+Z)/100→L
28: prt "Total
 A=",A
29: prt "Number
 of sightings=",
 C-1
30: prt "Distanc
 e to MLW=",L
31: plt 7000,
 500,1;lbl "Prof
 ile No.    Date
       ";pen
32: plt 7000,
 450,1;lbl "Tota
 l Area=",A;pen
33: plt 7000,
 400,1;lbl "Numb
 er of sightings
 =",C-1;pen
34: plt 7000,
 350,1;lbl "Dist
 ance to MLW ",L
35: ptyp
36: end
```

Figure 1. Program for beach profile evaluation.



Figure 2. Example of a beach profile plot.



Figure 3. Method of calculating area between beach surface and the mean low water base line.

Statements 0-7 are plotter commands that establish the graphics by setting the scale, labeling the abscissa and ordinate and marking the interval locators. The scale (o:scl) statement may be edited for the length of the profiles being measured. The scale, as illustrated in Fig. 1, will accommodate a profile 180 m (18,000 cm) in length with an initial elevation of no greater than 5.5 m (550 cm). Negative values or x-minimum and y-minimum values of the scale statement provide labeling space for the abscissa and ordinate.

Operation Three

The calculator displays "Initial Elevation?" and the operator keys in the elevation (cm) of the profile origin. After the initial elevation is entered, "continue" is pressed on the alphanumeric keyboard.

Statement 9 assigns a value of zero to x, the area variable (A) and the counter (C). Statement 10 commands the plotter to plot the initial elevation.

Operation Four

The calculator displays "Sighting interval in cm?" and the operator enters the sighting interval (spacing between profile sightings). The sighting interval (S) is entered when "Continue" is pressed.

Operation Five

The calculator displays "Change in elevation?" and the operator begins keying in each change in elevation along the profile transect. Every change in elevation is entered by depressing the "Continue" key.

Statements 13-15 plot elevation changes.

Statement 16 is a conditional statement for the final elevation change. When the elevation attains a negative value, the "go to" statement introduces the subroutine for completing the calculations.

Statements 17-19 calculate the area under the profile as each change in elevation is entered.

Statement 20 returns the program to the change in elevation inquiry for the next entry.

Statements 21-26 contain the subroutine for determining the area when the profile elevation is less than zero and where the profile crosses the zero elevation (MLW datum).

Statement 27 determines the distance along the profile surface, from the profile origin to mean low water.

Statements 28-34 are plot and print commands to display the calculated area, the distance to mean low water from the profile origin, and the number of sightings to mean low water.

Operation Six

The "ptyp" statement (35) enables the operator to print all alpha-numeric characters in the plotting area directly from the keyboard. The location and labeling of special features noted in the field may be completed in this mode.

Operation Seven

The operator presses "stop" when the plotted profile is complete.

APPLICATION AND ADVANTAGES

A prototype of this program was initially designed for use with the Shor-J rapid beach profiling method (Oertel, Chamberlain and Larsen, 1979). However, the original program did not have the flexibility to accommodate changes in sighting intervals or profile length without lengthy program editing. The program described above has no linear restrictions and computes profile lengths to mean low water. The program also is not

restricted by fixed data collection intervals. If, in the field, it is necessary to shorten or lengthen the sighting interval, this can be accommodated in the program by pressing "STOP"and then typing in "Cont 11" followed by the "EXECUTE" command. This returns the program to Statement 11 where the new sighting interval is entered and the "continue" key is pressed.

Any profiling method that records changes in elevation relative to a proceeding elevation along the profile transect may be used with the program. The staff and horizon method (Emery, 1961) produces data in a form that is readily suitable for the program.

REFERENCES

Davis, R. A. and W. T. Fox, 1971, Beach and nearshore dynamics in Eastern Lake Michigan: Western Mich. Univ. Tech. Report No. 4, p. 153.

Emery, K. O., 1961, A simple method of measuring beach profiles: Limnology and Oceanography, v. 6, p. 90-93.

Oertel, G. F., C. F. Chamberlain and M. C. Larsen, The Shor-J method for rapid beach profiling: Old Dominion Univ., Inst. of Oceanography Tech. Report No. 40, p. 11.